

Microsoft Excel – Match Index Offset



MATCH, INDEX and OFFSET

The MATCH function returns the relative position of an item in a range that matches a specified value. If there is no match then the #N/A error value is returned.

MATCH(lookup_value,lookup_array,match_type)

Where *Lookup_value* is the value to be matched, *Lookup_array* is the range of values and *Match_type* is the number -1, 0, or 1.

Match_type	Finds
1	The largest value that is less than or equal to <i>lookup_value</i> . <i>Lookup_array</i> must be in ascending order.
0	The first value that is exactly equal to <i>lookup_value</i> . <i>Lookup_array</i> can be in any order.
-1	The smallest value that is greater than or equal to <i>lookup_value</i> . <i>Lookup_array</i> must be in descending order.

Match is used to find the position of an item in a list and also to detect whether the match is available or not. Where VLOOKUP or HLOOKUP can not be used or you prefer not to tangle with their complexities, use a combination of MATCH and INDEX to look up a value. MATCH locates it and provides its index position and then INDEX retrieves it.

	B11	fx =MATCH(A11,C2:C9,0)		
	A	B	C	D
1	Population	GDP	Country	
2	266.7	7,600.0	United States	
3	125.7	5,100.0	Japan	
4	81.3	2,500.0	Germany	
5	58.4	1,600.0	France	
6	58.7	1,200.0	United Kingdom	
7	56.9	1,200.0	Italy	
8	164.2	883.3	Brazil	
9	1,230.0	636.7	China	
10				
11	Japan	2		

In this example, we want to look up the population of a country that is typed into cell A11. VLOOKUP is of no use here as the countries are located over to the right of the population figures.

The first task is to find the relative position of the country in the range of cells C2:C9. Match type value 0 is used as there is only one value to be found and the list of countries is not sorted

in any particular order.

That is the first bit, find the value. Next we have to retrieve the value out of the table.

To return the population of the matched country, the match calculation is placed as the row index argument inside a formula that uses

	A	B	C	D
1	Population	GDP	Country	
2	266.7	7,600.0	United States	
3	125.7	5,100.0	Japan	
4	81.3	2,500.0	Germany	
5	58.4	1,600.0	France	
6	58.7	1,200.0	United Kingdom	
7	56.9	1,200.0	Italy	
8	164.2	883.3	Brazil	
9	1,230.0	636.7	China	
10				
11	Japan	=INDEX(A2:C9,MATCH(A11,C2:C9,0),1)		
12		INDEX(array, row_num, [column_num])		

Microsoft Excel – Match Index Offset

the INDEX function. In the example, the range being indexed is A2:C9, MATCH provides the row index value and the column index value is 1 as the populations are listed in the first column of the range.

Alternatively, the formula could have been entered where the range being indexed was A2:A9, i.e. where the populations are listed and only the row index would be required as there is only one column in the area.

MATCH and INDEX formulas look more complicated than their VLOOKUP and HLOOKUP equivalents but are far more deliberate and less prone to error. They just have a two-step process, find something and then retrieve it.

An alternative referential method to using indexing is to point to the required value in the range using Excel's all-purpose pointer function, OFFSET. Once you have found the index of an item then there is no need to look it up as you know exactly where it is and all you have to do is point to it.

	A	B	C	D
1	Population	GDP	Country	
2	266.7	7,600.0	United States	
3	125.7	5,100.0	Japan	
4	81.3	2,500.0	Germany	
5	58.4	1,600.0	France	
6	58.7	1,200.0	United Kingdom	
7	56.9	1,200.0	Italy	
8	164.2	883.3	Brazil	
9	1,230.0	636.7	China	
10				
11	Japan	=OFFSET(A1,MATCH(A11,C2:C9,0),0)		
12		OFFSET(reference, rows, cols, [height], [width])		
13				

In the illustrated example, every position on the worksheet can be described as a column or row offset from cell reference A1. An offset value is the number of "moves" that you would have to make in order to get to one cell position from another. Offsets are given in the order Row, Column and are either positive or negative values; positive values are Down (row offsets) and to the

Right (column offsets) whereas negative values are Up (row) and to the Left (column).

MATCH calculates the index position of the country in column C, and can then be used as the row offset from A1 to locate the corresponding population in column A. The column offset is zero as it is in the same column. To return the GDP figure instead of the Population figure, the column offset value would be 1. Again, a two-step process; find something and then point to it.

Both the INDEX and OFFSET functions appear to be somewhat obscure when you examine the documentation on Excel's worksheet functions, they can only be understood in context. As you use them you come to realise that they are the most powerful and flexible look up functions of all and their implementation is usually rather more straightforward than VLOOKUP and HLOOKUP but you have to try them a few times...

INDEX(array,row_num,column_num)

Where *array* is a range of cells and *row_num* and *column_num* are index values.

Microsoft Excel – Match Index Offset

INDEX returns the value of an element in a table or an array, selected by the row and column number indices. There are a few different forms that the function takes but this is the most commonly used. Either the row or column index can be omitted if not required.

	A	B	C
1	Population	GDP	Country
2	266.7	7,600.0	United States
3	125.7	5,100.0	Japan
4	81.3	2,500.0	Germany
5	58.4	1,600.0	France
6	58.7	1,200.0	United Kingdom
7	56.9	1,200.0	Italy
8	164.2	883.3	Brazil
9	1,230.0	636.7	China
10			
11		United Kingdom	

INDEX is actually a very simple function and just processes an index value to retrieve the *n*th item from a range of cells. In the illustration, the *array* is given as C2:C9 and the *row_num* as 5, the formula returns the cell value at that position.

When both the row and column indices are given, INDEX returns the value in the cell at the intersection. If the *array* was A2:C9 and the index values then 5,2 it would return the GDP and the index values 5,1 the Population.

INDEX is a function rarely used in isolation; it is usually combined with another function.

OFFSET(reference,rows,cols,height,width)

Where *reference* is the starting point, *rows* and *cols* are the "moving away from the starting point" values and *height* and *width* are the area values. *Reference* is a cell reference and all the other arguments are numbers.

Excel Help describes this function thus, "Returns a reference to a range that is a specified number of rows and columns from a cell or range of cells". Which makes perfect sense if you have ever used the function but is as clear as mud if you have not. Perhaps we could just say that "it points to things". Not much clearer; you have to use it.

OFFSET points to other cells by using offset values from the starting point. If you point to one cell then you should leave out the area values as the last two arguments are the dimensions of the range that you are pointing at. Either use a zero value for the *rows*, *cols*, *height* and *width* arguments or use a comma separator as a positional marker to show that the argument value is not given.

	A	B	C	D	E
1	Item	Jan	Feb	Mar	Apr
2	Payroll	145,825	132,544	146,854	155,000
3	Insurance	550	550	550	550
4	Travelling	1,335	1,254	1,454	1,500
5	Consumables	5,985	6,524	7,254	7,500
6					
7	This Month				
8	Payroll	=OFFSET(A2,0,MONTH(NOW()))			
9	Insurance	OFFSET(reference, rows, cols, [height], [width])			
10	Travelling				

In the example, we want the formula in cell B8 to always show the relevant figure for the current calendar month by retrieving it from the table of monthly values.

Microsoft Excel – Match Index Offset

The expression MONTH(NOW()) returns the value of the current month, therefore we can start from cell A2 and offset by this number of months across row 2 to point to the relevant monthly figure.

There is no row offset required as all the numbers are along the same row. The row offset value is either given as zero or is omitted using commas as position markers:

=OFFSET(A2,,MONTH(NOW()))

This example is similar to the previous one but here we want to calculate the cumulative Year to Date total for each row, starting with the first month and continuing across to the current calendar month.

We need a SUM where the range of cells to add up is

	A	B	C	D	E
1	Item	Jan	Feb	Mar	Apr
2	Payroll	145,825	132,544	146,854	155,7
3	Insurance	550	550	550	5
4	Travelling	1,335	1,254	1,454	1,3
5	Consumables	5,985	6,524	7,254	7,8
6					
7	Year to Date				
8	Payroll	=SUM(OFFSET(B2,,,1,MONTH(NOW())))			
9	Insurance				

fixed at one end on column B, but the other end has to expand based on whatever month it is. Here we do not use the *rows* and *cols* arguments but instead the area values to refer to an area of cells which starts in column B and whose *height* is one cell and *width* is the value returned by MONTH(NOW())

The OFFSET function describes the dynamic range for the SUM function. Notice that there are three commas in the formula as placeholders for the argument values not given and there are a few closing parentheses to consider; one to close the NOW function, one for the MONTH function, one for the OFFSET function and, finally, one to close the SUM:

=SUM(OFFSET(B2,,,1,MONTH(NOW())))

